Matriisien käsittelyä

with(LinearAlgebra)

21.5.2012 Heikki Apiola

```
> restart:
> with(linalg):
  with(LinearAlgebra): # Muuta puolipisteeksi, jos haluat nähdä
  funktioluettelon.
```

Kirjasto linalg on vanhempi, versiosta 6 alkaen on uudempi. Pyrimme etupäässä käyttämään uutta, mutta joskus on mukavaa/tarpeellista soveltaa jotain vanhan kunnon **linalgin** funktiota. (Luultavasti jossain vaiheessa linalg poistuu, mutta se aika ei ole vielä käsillä.)

Koska uudet nimet ovat tosi pitkiä, kannattaa määritellä aliakset tähän tapaan:

```
> alias(rref=ReducedRowEchelonForm): alias(Diag=DiagonalMatrix):
> alias(ref=GaussianElimination):  # "row echelon form"
> alias(Det=Determinant,Id=IdentityMatrix):
```

Matriisien rakentelu ja laskutoimitukset

Rakentelu

Kaikki matriisit kannattaa ajatella muodostetuiksi ositetuista matriiseista.
Katsotaan seuraavassa muutamia esimerkkejä:

Maple	Matlab
> $\langle 1, 2, 3\rangle$; $\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$ **(2.1)**	>> [1;2;3]
Mapessa ei saa unohtaa pilkkuja, kuten Matlabissa > $\langle 1 \mid 2 \mid 3\rangle$; $\begin{bmatrix} 1 & 2 & 3 \end{bmatrix}$ **(2.2)**	>> [1 2 3] % tai [1,2,3]
> $\langle\langle 1, 3\rangle \mid \langle 2, 4\rangle\rangle$; $\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$ **(2.3)**	>> [1 2;3 4]

> $A := \langle\langle 1 \mid 2\rangle, \langle 3 \mid 4 \rangle \rangle;$ $$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \quad (2.4)$$ Vaihtoehtoinen tapa: > $A := Matrix([[1, 2], [3, 4]])$ $$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \quad (2.5)$$ > $\langle\langle A \mid \langle 1, 2\rangle\rangle, \langle a\|b\|c\rangle\rangle;$ $$\begin{bmatrix} 1 & 2 & 1 \\ 3 & 4 & 2 \\ a & b & c \end{bmatrix} \quad (2.6)$$ > Matriisin osia voi muutella (Matlabmaisesti). > $B := \langle\langle A \mid \langle 1, 2\rangle\rangle, \langle a\|b\|c\rangle\rangle;$ $$B := \begin{bmatrix} 1 & 2 & 1 \\ 3 & 4 & 2 \\ a & b & c \end{bmatrix} \quad (2.7)$$ > $B[1, 1..-1] := \langle x\|x\|x\rangle$: *# Muutetaan 1. rivi.* B; *# Katsotaan B.* $$\begin{bmatrix} x & x & x \\ 3 & 4 & 2 \\ a & b & c \end{bmatrix} \quad (2.8)$$ 1..-1 vastaa Matlabin pelkkää kaksoispistettä (:). Indeksi -1 vastaa Matlab:n **end**:iä	>> syms a b c >> B=[A [1;2] ;[a b c]] B = [1, 2, 1] [3, 4, 2] [a, b, c] >> syms x >> B(1,:)=[x x x] B = [x, x, x] [3, 4, 2] [a, b, c] Matlabissa voitaisiin kirjoittaa: B(1,1:end)=[x x x]

Voit vielä kokeilla seuraavia muuttamalla kaksoispisteet (:) puolipisteiksi (;).

```
> B[1,1..-2]:
> B[1..-1,2]:=<haa,hii,hoo>: B:
> B[2..3,[1,3]]:
> B[2..3,[1,3]]:=<<21,31>|<23,33>>: B:
```

Laskutoimitukset

Käydään läpi perusoperaatiot esimerkkien avulla.

```
> A := <<1,2>|<3,4>>;
```

```
B := <<a,b>|<c,d>>;
```

$$A := \begin{bmatrix} 1 & 3 \\ 2 & 4 \end{bmatrix}$$

$$B := \begin{bmatrix} a & c \\ b & d \end{bmatrix} \tag{2.2.1}$$

Kertolasku

```
> A . B;
```

$$\begin{bmatrix} a+3\,b & c+3\,d \\ 2\,a+4\,b & 2\,c+4\,d \end{bmatrix} \tag{2.2.2}$$

Transpoosi

```
> Transpose(A);
```

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \tag{2.2.3}$$

Gaussin eliminaatio rivioperaatioilla (käsin)

Aloitetaan restartista.

```
> restart:with(linalg):
  with(LinearAlgebra):
```

Esim:

```
> A := <<1,1,0,1>|<-1,1,2,-3>|<1,2,1,0>|<0,4,4,-4>|<0,1,3,-2>|<2,
  2,1,1>>;
```

$$A := \begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 \\ 1 & 1 & 2 & 4 & 1 & 2 \\ 0 & 2 & 1 & 4 & 3 & 1 \\ 1 & -3 & 0 & -4 & -2 & 1 \end{bmatrix} \tag{3.1}$$

```
> b:=<0,3,2,0>;
```

$$b := \begin{bmatrix} 0 \\ 3 \\ 2 \\ 0 \end{bmatrix} \tag{3.2}$$

```
> Ab:=<A|b>;
```

$$Ab := \begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 1 & 1 & 2 & 4 & 1 & 2 & 3 \\ 0 & 2 & 1 & 4 & 3 & 1 & 2 \\ 1 & -3 & 0 & -4 & -2 & 1 & 0 \end{bmatrix} \tag{3.3}$$

```
> Ab[2,1..-1]:=Ab[2,1..-1]-Ab[1,1..-1]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 2 & 1 & 4 & 3 & 1 & 2 \\ 1 & -3 & 0 & -4 & -2 & 1 & 0 \end{bmatrix} \tag{3.4}$$

```
> Ab[4,1..-1]:=Ab[4,1..-1]-Ab[1,1..-1]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 2 & 1 & 4 & 3 & 1 & 2 \\ 0 & -2 & -1 & -4 & -2 & -1 & 0 \end{bmatrix} \tag{3.5}$$

1. sarake nollattu. Siirrytään 2. sarakkeeseen.

```
> Ab[3,1..-1]:=Ab[3,1..-1]-Ab[2,1..-1]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 0 & 0 & 0 & 2 & 1 & -1 \\ 0 & -2 & -1 & -4 & -2 & -1 & 0 \end{bmatrix} \tag{3.6}$$

```
> Ab[4,1..-1]:=Ab[4,1..-1]+Ab[2,1..-1]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 0 & 0 & 0 & 2 & 1 & -1 \\ 0 & 0 & 0 & 0 & -1 & -1 & 3 \end{bmatrix} \tag{3.7}$$

2. sarake nollattu. Seuraava pivot on sarakkeessa 5.

```
> Ab[4,1..-1]:=Ab[4,1..-1]+(1/2)*Ab[3,1..-1]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 0 & 0 & 0 & 2 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{2} & \frac{5}{2} \end{bmatrix} \tag{3.8}$$

Ratkaise nyt "käsin"!

```
> x6:=solve(-1/2*x=5/2,x);
```

$$x6 := -5 \tag{3.9}$$

x5 ja x4 vapaita.

```
> x5:=solve(2*x+x6=-1,x);
```

$$x5 := 2 \tag{3.10}$$

Jne.   Kaikenkaikkiaan 2 vapaata parametria,x4 ja x3.

Jos haluttaisiin sarakepermutaatioita, niin näin:

```
> Ab[1..-1,[3,5]]:=Ab[1..-1,[5,3]]:Ab[1..-1,[4,6]]:=Ab[1..-1,[6,4]]: Ab;
```

$$\begin{bmatrix} 1 & -1 & 0 & 2 & 1 & 0 & 0 \\ 0 & 2 & 1 & 0 & 1 & 4 & 3 \\ 0 & 0 & 2 & 1 & 0 & 0 & -1 \\ 0 & 0 & 0 & -\frac{1}{2} & 0 & 0 & \frac{5}{2} \end{bmatrix} \quad (3.11)$$

Gaussin eliminaatio valmiilla funktioilla, aliakset ref ja rref

```
> restart:with(linalg):
  with(LinearAlgebra):
> alias(rref=ReducedRowEchelonForm):
> alias(ref=GaussianElimination):  # "row echelon form"
> A := <<1,1,0,1>|<-1,1,2,-3>|<1,2,1,0>|<0,4,4,-4>|<0,1,3,-2>|<2,2,1,1>>;
```

$$A := \begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 \\ 1 & 1 & 2 & 4 & 1 & 2 \\ 0 & 2 & 1 & 4 & 3 & 1 \\ 1 & -3 & 0 & -4 & -2 & 1 \end{bmatrix} \quad (4.1)$$

```
> b:=<0,3,2,0>:Ab:=<A|b>;
```

$$Ab := \begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 1 & 1 & 2 & 4 & 1 & 2 & 3 \\ 0 & 2 & 1 & 4 & 3 & 1 & 2 \\ 1 & -3 & 0 & -4 & -2 & 1 & 0 \end{bmatrix} \quad (4.2)$$

```
> E:=ref(Ab);
```

$$E := \begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 0 & 0 & 0 & 2 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{2} & \frac{5}{2} \end{bmatrix} \quad (4.3)$$

```
> rref(Ab);
```

(4.4)

$$\begin{bmatrix} 1 & 0 & \frac{3}{2} & 2 & 0 & 0 & \frac{21}{2} \\ 0 & 1 & \frac{1}{2} & 2 & 0 & 0 & \frac{1}{2} \\ 0 & 0 & 0 & 0 & 1 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 & 1 & -5 \end{bmatrix} \qquad (4.4)$$

Takaisinsijoitus ref-muodosta sekä ,LinearSolve

```
> E;
```

$$\begin{bmatrix} 1 & -1 & 1 & 0 & 0 & 2 & 0 \\ 0 & 2 & 1 & 4 & 1 & 0 & 3 \\ 0 & 0 & 0 & 0 & 2 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{2} & \frac{5}{2} \end{bmatrix} \qquad (5.1)$$

```
> LinearSolve(E[1..-1,1..-2],E[1..-1,-1]);
```

$$\begin{bmatrix} 9 + 3\,\_t0_2 + 4\,\_t0_4 \\ \_t0_2 \\ -2\,\_t0_2 + 1 - 4\,\_t0_4 \\ \_t0_4 \\ 2 \\ -5 \end{bmatrix} \qquad (5.2)$$

```
>
```